

Tom Warren and Lorena Matheus

Physics International Company
2700 Merced Street
San Leandro, CA 94577

Abstract

Nonlinear resistances occur in several different pulsed power applications. In the SPICE circuit simulation code family, dependent voltage and current sources can be used to model these varying resistors¹. Two problems of particular interest are a current-dependent load resistance, simulating plasma loads, and an action-dependent resistance, simulating high-voltage fuses. The current-dependent load resistance has been modeled using two current-controlled voltage sources. Two sources are required because the rising and falling portions of the load current waveform follow two different R(I) profiles. High-voltage fuses have been modeled using an action-dependent resistance. The manufacturer's data for the fuse resistance as a function of action is input into a table that is used to define the fuse resistance. This paper describes the development of the technique of modeling varying resistances with dependent sources.

Modeling Technique

The relationship between the voltage and current associated with a nonlinear resistance is given by:

$$V_I(t) = R(\alpha) i_I(t)$$

where the resistance, R, is dependent on some variable α . Modeling nonconstant resistances in SPICE used to be inconvenient at best. However, new features of recent versions of SPICE, such as PSpice, now allow much greater flexibility in specifying non-linear components.²

Resistance

Spice provides a standard format for modeling a constant resistance:

Rxxxxxx N1 N2 value

where:

- xxxxxx is an identifier which may be a mixture of numerals or letters.
- N1 and N2 are the two resistor nodes and
- "value" is the constant resistance value.

If the resistance that you are attempting to model is not constant, then we have to resort to a different modeling technique.

The equation for a resistance as a function of an independent variable, α , is given by Ohm's Law:

$$R(\alpha) = V(\alpha) / I(\alpha)$$

which may, of course, be rewritten as either

$$V(\alpha) = I(\alpha) * R(\alpha)$$

or

$$I(\alpha) = V(\alpha) / R(\alpha)$$

Technique

The general technique for modeling a varying resistance value is to utilize either a dependent voltage or a dependent current source as the

device model. For example, we may model a varying resistance by a current-dependent voltage source that has a value equal to the product of two currents. One of these two currents is the current through resistor. The other current is specially synthesized to generate a representation of the desired resistance with the proper dependence on the variable α .

Alternatively, we could model a varying resistor with a voltage-dependent current source that has a value equal to the product of the voltage across the resistor multiplied by another voltage. This second voltage is a source that generates a representation of the conductance with a proper dependence of the variable α .

Polynomials

Most versions of Spice allow the value of dependent sources to be represented by polynomials. The general form of the polynomial expression is given by:

$$P = P_0 + P_1 f_1 + P_2 f_2 + \dots + P_n f_n + P_{n+1} f_1 f_1 + P_{n+2} f_1 f_2 + \dots + P_{2n} f_1 f_n + P_{2n+1} f_2 f_2 + P_{2n+2} f_2 f_3 + \dots$$

where:

- $P_0, P_1, P_2, \dots$ are constant coefficients
- $f_1, f_2, \dots, f_n$ are the independent variables (node voltages or branch currents).

If a particular dependent source is dependent on a single parameter (be it a node voltage or branch current), then polynomial expression reduces to:

$$P = P_0 + P_1 f_1 + P_2 f_1^2 + P_3 f_1^3 + \dots$$

If a particular dependent source is dependent on two parameters (i.e., two node voltages or two branch currents), the polynomial expression is:

$$P = P_0 + P_1 f_1 + P_2 f_2 + P_3 f_1 f_1 + P_4 f_1 f_2 + P_5 f_1^2 f_2 + \dots$$

This is the expression we need for modeling our varying resistors.

Dependent Source Syntax

Four types of controlled sources exist: voltage-controlled voltage source, current-controlled voltage source, voltage-controlled current source, and current-controlled current source.

The syntax for the voltage-controlled voltage source is:

Exxxxxx N1 N2 Poly(y) NC1+ NC1- NC2+ NC2- ... P0 P1 P2 P3 ...

where:

- xxxxxx is an identifier which may be a mixture of numerals or letters,
- N1 and N2 are the two voltage source nodes (if positive current exits N1, the voltage source is delivering energy; if positive current exits N2, the voltage source is dissipating energy),
- y in Poly(y) indicates the number of controlling voltages,
- NC1+, NC1-, NC2+, NC2-, ... are the positive and negative terminals of the controlling node voltages, and

* This work was partially sponsored by FMC under subcontract number F42975 and by the U. S. Army Armament Research, Development and Engineering Center under contract number DAAA21-89-C-0132.

Report Documentation Page				Form Approved OMB No. 0704-0188	
Public reporting burden for the collection of information is estimated to average 1 hour per response, including the time for reviewing instructions, searching existing data sources, gathering and maintaining the data needed, and completing and reviewing the collection of information. Send comments regarding this burden estimate or any other aspect of this collection of information, including suggestions for reducing this burden, to Washington Headquarters Services, Directorate for Information Operations and Reports, 1215 Jefferson Davis Highway, Suite 1204, Arlington VA 22202-4302. Respondents should be aware that notwithstanding any other provision of law, no person shall be subject to a penalty for failing to comply with a collection of information if it does not display a currently valid OMB control number.					
1. REPORT DATE JUN 1991		2. REPORT TYPE N/A		3. DATES COVERED -	
4. TITLE AND SUBTITLE Modeling Non-Linear Pulsed Power Component Behaviors				5a. CONTRACT NUMBER	
				5b. GRANT NUMBER	
				5c. PROGRAM ELEMENT NUMBER	
6. AUTHOR(S)				5d. PROJECT NUMBER	
				5e. TASK NUMBER	
				5f. WORK UNIT NUMBER	
7. PERFORMING ORGANIZATION NAME(S) AND ADDRESS(ES) Physics International Company 2700 Merced Street San Leandro, CA 94577				8. PERFORMING ORGANIZATION REPORT NUMBER	
9. SPONSORING/MONITORING AGENCY NAME(S) AND ADDRESS(ES)				10. SPONSOR/MONITOR'S ACRONYM(S)	
				11. SPONSOR/MONITOR'S REPORT NUMBER(S)	
12. DISTRIBUTION/AVAILABILITY STATEMENT Approved for public release, distribution unlimited					
13. SUPPLEMENTARY NOTES See also ADM002371. 2013 IEEE Pulsed Power Conference, Digest of Technical Papers 1976-2013, and Abstracts of the 2013 IEEE International Conference on Plasma Science. Held in San Francisco, CA on 16-21 June 2013. U.S. Government or Federal Purpose Rights License.					
14. ABSTRACT Nonlinear resistances occur in several different pulsed power applications. In the SPICE circuit simulation code family, dependent voltage and current sources can be used to model these varying resistors. Two problems of particular interest are a current-dependent load resistance, simulating plasma loads, -and an action-dependent resistance, simulating high-voltage fuses. The current-dependent load resistance has been modeled using two current-controlled voltage sources. Two sources are required because the rising and falling portions of the load current waveform follow two different R(I) profiles. High-voltage fuses have been modeled using an action-dependent resistance. The manufacturer's data for the fuse resistance as a function of action is input into a table that is used to define the fuse resistance. This paper describes the development of the technique of modeling varying resistances with dependent sources.					
15. SUBJECT TERMS					
16. SECURITY CLASSIFICATION OF:			17. LIMITATION OF ABSTRACT SAR	18. NUMBER OF PAGES 6	19a. NAME OF RESPONSIBLE PERSON
a. REPORT unclassified	b. ABSTRACT unclassified	c. THIS PAGE unclassified			

- P0, P1, P2, P3, ... are the constant coefficients of the polynomial.

We only need to specify the polynomial coefficients up to the last one with nonzero value. For instance, if P0 = 32, P1 = 4.5, P2 = 0, P3 = 2, P4 = 0, and all other coefficients are 0, then we must specify, P0, P1, P2, and P3. Spice assumes that the remainder of the coefficients are zero.

An alternative syntax for the voltage-controlled voltage source is:

```
Exxxxxx N1 N2 NC+ NC- gain
```

where:

- N1 and N2 are the two voltage source nodes,
- NC+ and NC- are the positive and negative terminals of the controlling node voltages, and
- “gain” is a constant multiplicative factor.

In this case, the value of the voltage source is:

$$\text{Exxxxxx} = V(\text{NC+}, \text{NC-}) * \text{gain}$$

where:

- V(NC+,NC-) is the voltage difference between the nodes.

The syntax for the current-controlled current source is:

```
Fxxxxxx N1 N2 Poly(y) Vcntl1 Vcntl2 ... P0 P1 P2 P3 ...
```

where:

- xxxxxx is an identifier which may be a mixture of numerals or letters,
- N1 and N2 are the two current source nodes (by convention, positive current enters node 1 and exits node 2),
- y in Poly(y) indicates the number of controlling variables,
- Vcntl1, Vcntl2, ... are voltage sources through which the controlling currents are flowing (these sources may be zero-valued in which case they neither sink nor source energy but rather simply sense the value of the current through the branch), and
- P0, P1, P2, P3, ... are the constant coefficients of the polynomial.

An alternative syntax for the current-controlled current source is:

```
Fxxxxxx N1 N2 Vcntl gain
```

where:

- N1 and N2 are the two current source nodes,
- Vcntl is a voltage source through which the controlling current is flowing, and
- “gain” is a constant multiplicative factor.

In this case, the value of the current source is:

$$\text{Fxxxxxx} = I(\text{Vcntl}) * \text{gain}$$

where:

- I(Vcntl) is the current through Vcntl.

The syntax for the voltage-controlled current source is:

```
Gxxxxxx N1 N2 Poly(y) NC1+ NC1- NC2+ NC2- ... P0 P1 P2 P3 ...
```

or alternatively:

```
Gxxxxxx N1 N2 NC+ NC- transconductance
```

where the parameters are similar to those given above for the voltage-controlled voltage source.

The syntax for the current-controlled voltage source is:

```
Hxxxxxx N1 N2 Poly(y) Vcntl1 Vcntl2 ... P0 P1 P2 P3 ...
```

or alternatively:

```
Hxxxxxx N1 N2 Vcntl transresistance
```

where the parameters are similar to those given above for the current-controlled current source.

Switching Example

Suppose we want to model a closing switch, between nodes 26 and 28, in which the resistance decreases from an open circuit (1 kΩ) to 1 mΩ over a 10 ns period beginning at 120 ns into the simulation. The Spice implementation of such a switch is:

```
* Use a voltage with value 0 to measure the load current.
VCVR 26 27 0.
* Implement the closing switch with a current-controlled voltage source.
HSWITCH 27 28 POLY(2) VCVR VRESISTANCE 0. 0. 0. 1.
* Put in a current source to generate the proper resistance
* Drive this current through the sensing voltage source
IRESISTANCE 0 98 PWL(0. 1000. 120.N 1000. 121.N 202. 122.N
+ 40.8 123.N 8.23 124.N 1.66 125.N .336 126.N .0687 127.N
+ .0148 128.N .00376 129.N .00156 130.N .001 1. .001)
RRES 98 99 1.0
VRESISTANCE 99 0 0.
```

The switch current sensing voltage source, VCVR, and the current-controlled voltage source, HSWITCH, are placed together between the switch nodes 26 and 28. The auxiliary loop formed by IRESISTANCE, RRES, and VRESISTANCE generate and sense the resistance profile. PWL in IRESISTANCE is a standard SPICE piece-wise linear statement in which each pair of numbers represents a time and a corresponding current. The value of GSWITCH is given by:

$$\begin{aligned} \text{HSWITCH} &= 0 + 0 * I(\text{VCVR}) + \\ &\quad 0 * I(\text{VRESISTANCE}) + \\ &\quad 0 * I(\text{VCVR})^2 + 1 * I(\text{VCVR}) * I(\text{VRESISTANCE}) \\ \text{HSWITCH} &= I(\text{VCVR}) * I(\text{VRESISTANCE}) \end{aligned}$$

In other words, the value of the voltage between nodes 27 and 28 is equal to the current flowing into node 27 multiplied by the current flowing into node 99. But we have chosen IRESISTANCE to force the current flowing into node 99 to be exactly equal to the desired resistance as a function of time.

Spice Extensions

The analog behavioral modeling modules of newer versions of PSpice have implemented extensions to the controlled sources that make this modeling process easier and more straight forward.² These new features include using both mathematical expressions and look-up tables to assign values to dependent sources. However, these new features only apply to voltage-controlled voltage sources (E) and voltage-controlled current sources (G).

Mathematical Expression Syntax

The syntax for using mathematical expressions for assigning value to a voltage-controlled voltage source is:

```
Exxxxxx N+ N- VALUE = (expression)
```

Examples include:

```
ESQROOT 7 4 VALUE = (3*SQRT(V(5,9)))
```

```
EVAC 77 34 VALUE = (5*SIN(6.28/.0001*TIME))
```

EPWR 13 31 VALUE = (V(99)*I(VCVR))

The syntax for using mathematical expressions for assigning value to a voltage-controlled current source is:

Gxxxxxx N+ N- VALUE = (expression)

where:

- xxxxxx is an identifier
- N1 and N2 are the two current source nodes, and
- “(expression)” is a valid mathematical expression involving constants, node voltages, source currents, and time, enclosed in parenthesis.

Examples include:

GDIVIDE 7 4 VALUE = ((V(5,9)/V(21,22))

GLOG 47 44 VALUE = (15*LOG(V(54,66)/235))

GFULL 3 1 VALUE = (PWR(V(99),1.33))

* voltage-controlled current source

GSWITCH 27 28 VALUE = V(27,28)/V(98,0)

Look-up Table Syntax

The syntax for using a look-up table for assigning value to a voltage-controlled voltage source is:

EXxxxxx N+ N- TABLE {expression} = (x1,y1) (x2,y2) (x3,y3)...(xn,yn)

where:

- xxxxxx is an identifier.
- N1 and N2 are the two voltage source nodes.
- “expression” is a valid mathematical expression.
- (x1,y1), etc., are x-y pairs of input and output values.

The syntax for the voltage-controlled current source is the same as above except the leading E is replaced by G.

The switching example given above can be implemented using a look-up table.

* Implement the closing switch with a
HSWITCH 27 28 VALUE = I(VCVR) voltage-controlled current source
V(98,0)
* Put in a current source to generate the proper resistance
GRESISTANCE 0 98 TABLE {TIME} (0. 1000.) (120.N 1000.)
+(121.N 202.) (122.N 40.8) (123.N 8.23) (124.N 1.66)
+(125.N .336) (126.N .0687) (127.N .0148) (128.N .00376)
+(129.N .00156) (130.N .001) (1. .001)
RRES 98 0 1.0

Modeling Nonlinear Current-Dependent Resistance

Current-Dependent Resistances

Suppose we wish to model some plasma load in which the resistance of the load is dependent on the conducted current as shown in Figure 1. The load has one current dependence during the leading edge of the current pulse (upper curve) but a completely different current dependence as the current is decreasing on the trailing edge (lower curve). We must make some assumptions here. We will assume that the system always starts out at $t = 0$ with no current flowing in the load. Secondly, we assume that the current increases monotonically to peak, then decreases monotonically. Thirdly, we assume that the current always reaches peak where the two current profiles have the same resistance values. Otherwise, the current would necessarily follow a different profile than either of the two given.




Figure 1. Non-linear resistance profile of a current dependent load.

Modeling Current Dependent Resistances

Because we have a multivalued function, we need to generate two values of resistance, one for the leading edge current waveform, and a second for the trailing edge waveform. Then, to compute the load voltage, we use the leading edge resistance profile during the rising current, then at peak, switch out the leading edge profile and switch in the trailing edge resistance profile for the decreasing current. However, in practice, doing this switching at peak leads to a “noisy” response.

A second method to achieving the same result is to use the leading edge resistance profile during the rising current, then at peak and during the decreasing current, subtract from the leading edge the amount required to trace out the trailing edge. This subtracted amount must be equal to the DIFFERENCE between the leading edge and trailing edge profiles.

A current-dependent resistance is most easily implemented using a current-controlled voltage source. Depending on the nature of the current dependence, the voltage value may be specified by either a look-up table or a polynomial that has been fit to the resistance profiles. We have chosen polynomials because both resistance profiles given in Figure 1 can be fit nicely with fourth order polynomials. However, what we actually need is one polynomial for the leading edge resistance, and another for the difference between the leading and trailing edge. The polynomials that specify these resistances are given by:

$$R_{\text{leading}}(I) = 1.4974 \times 10^{-23} * I^4 - 1.3212 \times 10^{-17} * I^3 + 5.5255 \times 10^{-12} * I^2 - 1.3532 \times 10^{-6} * I + 0.16135$$

$$R_{\text{difference}}(I) = 1.0409 \times 10^{-23} * I^4 - 8.5892 \times 10^{-18} * I^3 + 3.1695 \times 10^{-12} * I^2 - 6.2340 \times 10^{-7} * I + 4.9959 \times 10^{-2}$$

Now, recognizing that the voltage is equal to the product of the current and the resistance, which is already a function of the current, we may specify the values of the current-dependent voltage sources to be:

$$HLOADL = 1.4974 \times 10^{-23} * I^5 - 1.3212 \times 10^{-17} * I^4 + 5.5255 \times 10^{-12} * I^3 - 1.3532 \times 10^{-6} * I^2 + 0.16135 * I$$

$$HLOADD = 1.0409 \times 10^{-24} * I^5 - 8.5892 \times 10^{-18} * I^4 + 3.1695 \times 10^{-12} * I^3 - 6.2340 \times 10^{-7} * I^2 + 4.9959 \times 10^{-2} * I$$

Combining the Responses

We now must combine these two voltage sources in such a way that the resultant voltage is equal to HLOADL during the rising current and HLOADL-HLOADD during the falling current. To accomplish this, we use a series combination of HLOADL and a second voltage source, ELOADT, where:

ELOADT = -HLOADD * Vstep

where:

- Vstep = 0 for 0 ≤ t < tpk
- Vstep = 1 for t > tpk
- tpk = time of current peak

The controlling “step function” Vstep is generated by a constant voltage source switched into a constant resistance, where the voltage controlled switch “turns on” at the current peak. The controlling voltage for this switch is generated in a sample-and-hold circuit consisting of a current-dependent voltage source charging a capacitor through a series resistor.

The Complete Model




Figure 2. Example circuit for two capacitors driving a non-linear current-dependent load resistance.

Figure 2 shows an example circuit of two capacitors driving a nonlinear current-dependent load. The PSpice input file for this example is:

```
Non-linear current dependent load model
* Driver model
C1      4      0      20.E-3      IC=12.8E3
RDUMB   4      0      1.E12
L1      4      3      72.E-6      IC=0.
SSW     3      5      190      0 SW1
L2      6      5      54.E-6      IC=0.
C2      6      0      15.E-3      IC=11.E3
RDUMB   6      0      1.E12
* CURRENT DEPENDENT LOAD *
VCVR    3      2      0.
HLOADL  2      1      POLY(1) VCVR 0. .16135 -1.3532E-6
+ 5.5255E-12 -1.3212E-17 1.4974E-23
ELOADT  0      1      VALUE = {V(291,0) * V(390,0)}
*
* SWITCH CONTROL SOURCE *
VCNTL1  190     0      PWL(0, 0. 1.E-3 0. 1.01E-3 1. 1. 1.)
RCNTL1  190     0      1.
*
* ENABLING CIRCUIT THAT "TURNS ON" THE DIFFERENCE
*SOURCE AT CURRENT PEAK *
VENABLE 290     0      1.0
SENABLE  290    291    491      0 SW2
RENABLE  291     0      1.
*
* CURRENT CONTROLLED VOLTAGE SOURCE TO
* SYNTHESIZE DIFFERENCE RESISTANCE *
HLOADD  390     0      POLY(1) VCVR 0. 4.9959E-2 -6.2340E-7
+ 3.1695E-12 -8.5892E-18 1.0409E-23
RLOADD  390     0      1.
*
* SAMPLE AND HOLD CIRCUIT TO SENSE CURRENT PEAK *
```

```
HSAMP    490     0      VCVR    1.
DSAMP    490    491      D1
CSAMP    491     0      1.      IC=0.
RSAMP    491     0      1.E12
*
* PSPICE CONTROL *
.MODEL SW1 VSWITCH (RON=.001)
.MODEL SW2 VSWITCH (RON=.0001, VON=200E3, VOFF=198E3)
.MODEL D1 D
.PROBE
.TRAN          10E-6  4.2E-3  UIC
.END
```

The load voltage and current for this example is shown in Figures 3 and 4, respectively. The resistance as a function of both time and current is shown in Figures 5 and 6, respectively. The hysteresis of the load resistance is clearly demonstrated in Figure 6.




Figure 3. Voltage across the current-dependent resistance.




Figure 4. Current through the current-dependent resistance.




Figure 5. Resistance as a function of time for the current-dependent load.




Figure 6. Load resistance as a function of load current.

Modeling High-Voltage Fuses

Fuse Operation

High-voltage fuses are used in capacitor banks to isolate faulted capacitors from the remainder of the bank. When a capacitor fails short, the current through the fuse on the faulted capacitor increases dramatically as the rest of the capacitors discharge into the short. As the fuse action (integral of the square of the current) builds, the resistance of the fuse increases. Figure 7 shows the manufacturer's data for fuse resistivity as a function of specific action.³ This chart can be coupled with fuse design information to generate the resistance of the fuse as a function of action, $R(\text{act})$.




Figure 7. Resistivity as a function of specific action.

Fuse Modeling

The fuse is most accurately modeled as an action-dependent resistance. We generate the dependent resistance as usual with a voltage-controlled current source of the form:

$$GFUSE\ 614\ 615\ VALUE = (V(614,615)/V(810,0))$$

where $V(810,0)$ is synthesized to be equal to the fuse resistance. How do we construct the fuse resistance?

The fuse resistance as a function of action data is input into a PSpice look-up table where the input value is the fuse action and the output value is the resistance.

$$ERESIST10\ 810\ 0\ TABLE\ \{V(0,710)/1E6\} = (0.002,0.025)$$

Now to look up the resistance in this table we need to construct $V(0,710)$ to be equal to the fuse action.

Since action is the integral of the fuse current squared:

$$\text{action} = \int I_{\text{fuse}}^2 dt$$

So we use a current dependent current source to generate the current squared.

$$VCVRFUS1\ 615,\ 215.0. \\ FSQUARE\ 710\ 0\ POLY(1)\ VCVRFUS1\ 0.0.1.$$

To integrate the value we can recognize:

$$V_c = 1/c \int I dt$$

So if we drive the a current through a capacitor, the capacitor voltage will be equal to the integral of the current.

$$CINTEG\ 710\ 0\ 1.\ IC=0.$$

We use the voltage across the capacitor as the input parameter to the look-up table.

Model Results

Figure 8 shows an example capacitor fusing circuit. The PSpice input deck for this example is:




Figure 8. Example capacitor fusing circuit.

```
FUSE MODELED BY ACTION DEPENDENT RESISTANCE
* 250 KJ CAPACITOR BANK MODEL *
* NINE CAPACITORS IN PARALLEL
C9          211    0          3.72E-3    IC=11E3
RSC9        211    0          1.E9
LC9         211    212        11.E-9      IC=0.
RC9         212    213        0.67E-3
LFUSE9      213    214        28.E-9
RFUSE9      214    215        1.4E-3
* A SINGLE CAPACITOR AT WHICH THE FAULT OCCURS
C1          611    0          413E-6      IC=11E3
LC1         611    612        100E-9      IC=0.
RC1         612    613        6.E-3
LFUSE1      613    614        250.E-9
***** Automatic fuse opening test circuit *****
* Fuse modeled by voltage dependent current source
GFUSE1      614    615        VALUE = {V(614,615)/(V(810,0))}
VCVRFUS1    615    215        0.
* Auxiliary circuit to calculate the action
FSQUARE     710    0          POLY(1) VCVRFUS1 0. 0. 1.
CINTEG      710    0          1.          IC=0
Rresist     710    0          10.0e9
* Voltage Source to produce the fuse resistance as a function of action
ERESIST10   810    0          TABLE {V(0,710)/1E6} =
+(0.00155045,0.025) (0.00775659,0.025) (0.01240429,0.025)
+(0.01602328,0.025) (0.02067714,0.025) (0.024817,0.025)
+(0.02868928,0.025) (0.03282061,0.025) (0.03618159,0.025)
+(0.04134917,0.025) (0.04678433,0.025) (0.05015042,0.025)
+(0.05376022,0.025) (0.0597118,0.025) (0.06487938,0.025)
+(0.07470495,0.025) (0.07936463,0.025) (0.08427145,0.025)
+(0.08863628,0.025) (0.0927625,0.025)
+(0.09924073,0.03674158) (0.10391234,0.03774707) (0.10907652,0.03885762)
+(0.11421001,0.04510068) (0.1188714,0.05123401) (0.12273686,0.06232785)
+(0.12685456,0.07347227) (0.13150402,0.07962972) (0.1351036,0.09576268)
+(0.14054729,0.10209757) (0.14569953,0.13400549) (0.149611,0.15019293)
+(0.15316968,0.17669207) (0.15910081,0.19333088) (0.1650166,0.21516835)
+(0.17021145,0.23169043) (0.17539437,0.25325552) (0.18127437,0.27503851)
+(0.18671123,0.30701648) (0.1905801,0.3437495) (0.19393767,0.38564227)
+(0.19726114,0.46331082) (0.19874392,0.54560991) (0.19974949,0.61763621)
+(0.20073801,0.70514174) (0.20165836,0.79263949) (0.2024594,0.89552309)
+(0.20266392,1.30098718) (0.20286844,1.40371513) (0.20307297,1.62442456)
+(0.20327749,1.75797089) (0.20337975,1.90147872) (0.20377175,2.20927344)
+(0.20387401,2.52228238) (0.20552723,5.15593119) (0.2200,1000.0)
*
Rdum810     810    0          1.0e6
*
```

* CAPACITOR FAULT ARC *

```
LARC      611      600      50E-9
RARC      600      0       1.5E-3
.PROBE
.OPTIONS LIMPTS=1000 RELTOL=.001 TRTOL=2.
+ ABSTOL=1.E-6 ITL5=0 VNTOL=1.E-3 CHGTOL=1.E-8
.TRAN 5.US 5.MS UIC
.END
```

Figures 9 and 10 show the fuse current and action, respectively, as functions of time. The total fuse action reaches 205,000 A²-s, which is within 10% of the quoted "total let-through" action of 220,000 A²-s.




Figure 9. Fuse current as a function of time.




Figure 10. Fuse action as a function of time.

Summary

PSpice, available in PC and Macintosh versions, can now be used to model varying resistances that have a wide variety of functional dependencies. These problems have historically required larger mainframes and powerful codes such as NET-2 and SCEPTRE. The general technique involves synthesizing resistors from dependent voltage and current sources. A current dependent load resistance has been modeled using two current-controlled voltage sources. Two sources are required because the rising and falling portions of the load current waveform follow two different R(I) profiles. High-voltage fuses have been modeled using an action-dependent resistance. The manufacturer's data for the fuse resistance as a function of action is input into a table that is used to define the fuse resistance.

References

- [1] SPICE Version 2G User's Guide, A. Vladimirescu, K. Zhang, A. R. Newton, D. O. Pederson, A. Snagiovanni-Vincentelli, 10 August 1981, Department of Electrical Engineering and Computer Sciences, University of California, Berkeley, CA, 94720.
- [2] PSpice, MicroSim Corporation, Version 4.03, January 1990, 20 Fairbanks, Irvine, CA, 92718.
- [3] "Pulsed Discharge Fuses for Capacitor Bank Protection," N. C. Jaitly, G. Schofield, and M. Briscoe, Proceedings of the 7th IEEE Pulsed Power Conference, Monterey, CA, June 1989, pp. 408 - 411.